APPLICATION NOTE 238-1

ULTRA LOW CURRENT SEMICONDUCTOR DC PARAMETER MEASUREMENT SYSTEM USING HP4140B

HEWLETT PACKARD

INTRODUCTION

Process engineers and process design engineers have long wanted an inexpensive, easy-to-use, automatic measuring system capable of quick and accurate characterization of semiconductor devices, particularly field effect and bipolar transistors. They have especially wanted a system capable of measuring DC parameters at ultra-low current levels, in the picoampere range.

Systems presently used for characterization of semiconductor devices are not only expensive but also difficult to use, requiring a great deal of time, effort, and expertise on the part of process engineers and process design engineers to set-up and maintain.

This application note describes two applications for the Model 4140B pA Meter/DC Voltage Source, configured into just such an automatic measuring system. The first application describes the measurement of the h_{FE}-I_C characteristics of a bipolar transistor; the second describes the measurement of a field effect transistor's (FET) transconductance, g_m. In both applications, measurement is made down to the 10^{-15} A range.

The measuring system used in both applications consists of the 4140B and an HP desktop computer (the Model 9845T or the inexpensive Model 85A). Both computers have a graphics CRT for quick evaluation of device characteristics, and both are equipped with a built-in thermal printer for accurate, inexpensive hard-copies of characteristics plots.



Fig. 1 Automatic Semiconductor DC Parameter Measurement System

(Note : The photograph on the front-cover shows the system with the HP85A Desktop Computer)

CONTENTS

ABOUT THE SYSTEM

In order to realize a truly automatic system for measuring the low current (10^{-15}A) DC parameters of semiconductor devices such as bipolar and field effect transistors, a digital picoammeter with the following capabilities is required:

(1) Measurement resolution of 0.001 pA.

(2) HP-IB compatible.

(3) Two programmable voltage sources.

(4) Rejection of externally generated noise, and prevention of undersired leakage current from flowing into the meter.

(5) Timing control between applied voltage changes and the actual measurement.

Until recently, though, no picoammeter had all of these capabilities. Thus, a costly, complicated multi-instrument system had to be configured, presenting engineers with a multitude of interfacing, timing, and noise problems. Also, resolution and accuracy were insufficient.

The HP Model 4140B pA Meter/DC Voltage Source, minimally systemized with either the Model 9845T Desktop Computer or the 85A Personal Computer, provides a complete solution.

Figure 2 shows a block diagram of the system. Although simple, this system has some really powerfull capabilities: I–V characteristics measurements over the current range of 10^{-15}A to 10^{-2}A, calculation of essential parameters for semiconductor evaluation, CRT and hard-copy plots of measured characteristics, and more.

The system can be used to evaluate the semiconductor manufacturing process, as well as the semiconductor itself. Device characteristics are plotted on the controller's CRT for quick evaluation, and, if desired, clear and accurate hard-copies of the displayed plot can be made. Also, multi-color, high resolution plots are possible by simply connecting an HP plotter.



Fig. 2 System Block Diagram

APPLICATION : h_{FE}–I_C Characteristics Measurement

In many applications of bipolar transistors, it is important not only that h_{FE} (DC current gain) be large but also that it not vary with changes in current levels. It is evident that in the absence of recombination within the emitter-base space-charge region, the current gain is independent of the collector current. The larger the recombination rate, the more the current gain drops at low current levels.

Such recombination can also cause noise current inside the transistor, so that the measurement of very low level current is useful in the development of low noise transistors. In measuring h_{FE} with the 4140B system, the V_B voltage source supplies the voltage between collector and base, which is set to zero for making the collector voltage equal to base voltage. The other voltage source V_A (_┌‾) of the 4140B is used to drive the voltage between emitter and base (same potential as collector). Because the pA meter of the 4140B is floating, it can be inserted either between the collector and the collector voltage supply or between the base and ground. Consequently, by making only one connection change, both I_C–V_{BE} and I_B–V_{BE} characteristics measurements can be made using the same program. The sample's h_{FE} is then calculated from these measured values as:

$$h_{FE} = \left.\frac{I_C}{I_B}\right|_{V_{BE}}$$

Sample plots are shown in Figures 5 and 6.
The connection used for the h_{FE} measurement is shown in Figure 3. The program listing is given in Appendix A.

APPLICATION : Transconductance Measurement of FET

The gate impedance, i.e., the input impedance of an FET (field-effect transistor) is very high. This high input impedance is one of the distinguishing characteristics of field-effect transistors as compared to junction transistors. At room-temperature, leakage currents of a reverse biased silicon p-n junction prior to breakdown are of the order of picoamps to nanoamps. However, surface effects can lead to drastic increases in junction leakage currents, and, thus, degrade the high input impedance of the transistor. Therefore, subthreshold characteristics measurements are useful for evaluation of the diffusion or ion doping process where impurity contamination may occur, producing leakage current via the surface or through the depletion region.

Transconductance, g_m, is one of the essential parameters to evaluate the manufacturing process of J-FETs or MOS FETs. The g_m value at some change of gate voltage or g_m–V_{GS} characteristics can be derived from I_D–V_{GS} (drain current vs gate voltage).

This I_D–V_{GS} characteristic must be measured in the subthreshold region (down to 10^{-15} A) of drain current to the current (up to 10^{-2} A) where the FET is completely turned on. For such a wide (11 decade) range, the g_m value is calculated as follows:

$$g_m = \frac{\partial I_D}{\partial V_{GS}} = \left.\frac{\Delta I_D}{\Delta V_{GS}}\right|_{V_{DS}}$$

The system using the HP 4140B measures current from the subthreshold region to turn-on (see Figure 7), and calculates the g_m–V_{GS} characteristics as shown in Figure 8. Figure 4 is the connection for measurement of I_D–V_{GS} characteristics. The program listing is given in Appendix B.



(a) I_C-V_{BE} measurement (b) I_B-V_{BE} measurement

Fig. 3 Connection for I_C, I_B–V_{BE} and h_{FE}–I_C Characteristics Measurements



Fig. 4 Connection for I_D–V_{GS} and g_m–V_{GS} Characteristics Measurements



Fig. 5



Fig. 7



Fig. 6



Fig. 8

Note) 1E ± N = 1 x $10^{\pm N}$

(Refer the program listings in Appendix A and B)

SUMMARY : The 4140B System Contribution

The system described in this application note is useful:

- To measure model parameters for circuit simulation in semiconductor device development/Integrated Circuit design,
- To measure device DC parameters for process control and evaluate other control parameters for maintaining high reliability.

And because the system is based on a pA Meter capable of measuring current-voltage characteristics down to 10^{-15} A region, it is well suited for quality control applications, such as gate leakage current measurements essential in failure analysis of the semiconductor process.

REFERENCES :

1. Grove, A. S. 1967. Physics and Technology of Semiconductor Devices.
2. Noguchi, H. 1979. Hewlett-Packard Journal, Vol. 30, No. 12, pp. 10 - 19.
3. Hewlett-Packard Application Note 238 "Semiconductor Measurements with the HP4140B Picoammeter/DC Voltage Source."

4140B SPECIFICATIONS

MEASUREMENT FUNCTIONS: I, I-V and C-V
CURRENT RANGE: 0.001 x 10^{-12}A – 1.999 x 10^{-2}A
DISPLAYS: 3-1/2 digits
BASIC ACCURACY: 0.5%
MEASUREMENT TIME: Approx. 4ms – 2.56s
CALCULATED CAPACITANCE RANGE: 0.1pF – 1999pF
VOLTAGE SOURCE (V_A and V_B): 0.00 to ±100.0V

Function	V_A	V_B
I		
I-V		
C-V		

RAMP RATE: 0.001V/s – 1.000V/s
SWEEP MODE: Auto/Manual (Pause)

APPENDIX A: "h_{FE}-I_C" Program Listing for the HP85A

For DUT connection, refer to Figure 3.
One connection change is all that is necessary for $I_C - V_{BE}$ and $I_B - V_{BE}$ measurements.

```
10 ! *** Hfe-Ic PROGRAM ***
20 ! .."HFE"... MAR.12.1981
30 OPTION BASE 1
40 DIM D$[30],A(200),C(200),V(2
   00),V0(200)
50 M=717
60 DISP "4140B Address Code ?(7
   17)"
70 INPUT M
80 GOSUB 170 ! Meas-setting
90 GOSUB 290 ! Measurement
100 GOSUB 540 ! Graphic-output
110 END
120 !
130 IMAGE 2A,M3DZ.2D,","
140 IMAGE MZ.DD," V ",MZ.DDDe,"
    A"
150 IMAGE MZe
160 !
170 ! *** Meas-setting ***
180 DISP "DUT=? (Up to 10 charac
    ters)"
190 INPUT D$
200 DISP "Input for Vbe sweep;
              START V? , STOP V?
    ,  STEP V?, STEP DELAY(sec)
    ?, Vcc?"
210 INPUT V1,V2,V3,T1,V0
220 N1=ABS((V2-V1)/V3)
230 IF FP(N1)#0 THEN N2=INT(N1)+
    2 ! N2 is # of STEP
240 IF FP(N1)=0 THEN N2=INT(N1)+
    1 ! N2 is # of STEP
250 REDIM A(N2),C(N2),V(N2),V0(N
    2)
260 RETURN
270 !
280 ! *** Measurement ***
290 ! *** Ic-Vbe Measurement ***
300 DISP "Connect DUT for Ic-Vbe
    , C:HI, B:GND, E:Va, LO:Vb.
    Then Press 'CONT'"
310 PAUSE
320 DISP "Measuring"
330 OUTPUT M ;"F2RA1I1J1T2A3B1L3
    M3"
340 OUTPUT M USING 130 ; "PS",V1
    ,"PT",V2,"PE",V3,"PD",T1,"PB
    ",V0
350 OUTPUT M ;"W1"
360 FOR I=1 TO N2
370 ENTER M ; A(I),V(I)
380 DISP USING 140 ; V(I),A(I)
390 NEXT I
400 ! *** Ib-Vbe Measurement ***
410 DISP "Change connection for
    Ib-Vbe, C:GND, B:HI, E:Va, L
    O:Vb"
420 PAUSE
430 DISP "Measuring"
440 OUTPUT M ;"F2RA1I1J1T2A3B1L3
    M3"
450 OUTPUT M USING 130 ; "PS",V1
    ,"PT",V2,"PE",V3,"PD",T1,"PB
    ",V0
460 OUTPUT M ;"W1"
470 FOR I=1 TO N2
480 ENTER M ; C(I),V0(I)
490 DISP USING 140 ; V0(I),C(I)
500 NEXT I
510 RETURN
520 !
530 ! *** Graphic-output ***
540 M2=0
550 M1=100
560 FOR I=1 TO N2
570 M2=MAX(M2,ABS(A(I)))
580 M1=MIN(M1,ABS(A(I)))
590 NEXT I
600 GCLEAR
610 GRAPH
620 LOCATE 30,110,25,88
630 PEN 1
640 M1=1 E-15
650 M2=.01
660 SCALE 0,V2,LGT(M1),LGT(M2)
670 AXES .2,1,0,LGT(M1)
680 DEG
690 X=V2-V1
700 Y=LGT(M2)-LGT(M1)
710 Y1=LGT(M1)
720 MOVE V1+X*.55,Y1+Y*1.15
730 LORG 5
740 LABEL "Ic,Ib-Vbe Characteris
    tics"
750 MOVE V1+X*.25,Y1+Y*1.05
760 FXD 2
770 LABEL "DUT=";D$
780 MOVE V1+X,Y1+Y*1.05
790 IMAGE "Vcb=",MZ.DD,"V"
800 LABEL USING 790 ; V0
810 MOVE V1-X*.3,Y1+Y*.55
820 LDIR 90
830 LABEL "Ic,Ib (A)"
840 LDIR 0
850 MOVE V1+X*.31,Y1-Y*.2
860 LABEL "HP 4140B"
870 MOVE V1+X*.99,Y1-Y*.2
880 LABEL "Vbe (V)"
890 PEN 1
900 ! *** Vbe Axis unit plot ***
910 CSIZE 3
920 LDIR PI/2
930 LORG 6
940 FOR V=0 TO V2 STEP  1*SGN(V2
    )
950 MOVE V,LGT(M1)-.2
960 FXD 1
970 LABEL V
980 NEXT V
990 ! ***I Axis Unit Plot ***
1000 CSIZE 3
1010 LDIR 0
1020 LORG 8
1030 FOR I=LGT(M1) TO LGT(M2) ST
     EP 1
1040 MOVE 0-X*.01,I
1050 LABEL USING 150 ; 10^I
1060 NEXT I
1070 PEN 1
1080 LINETYPE 1
1090 FOR I=1 TO N2
1100 PLOT (I*V3-V3)*SGN(V2),LGT(
     ABS(A(I)))
1110 IF I*V3-V3>V2*SGN(V2) THEN
     1130
1120 NEXT I
1130 PENUP
1140 LINETYPE 2
```

```
1150 FOR I=1 TO N2
1160 PLOT (I*V3-V3)*SGN(V2),LGT(
     ABS(C(I)))
1170 IF I*V3-V3>V2*SGN(V2) THEN
     1230
1180 NEXT I
1190 LINETYPE 1
1200 COPY
1210 DISP "PRESS 'CONT'"
1220 PAUSE
1230 ! *** Hfe Axes ***
1240 H2=1000
1250 H1=.01
1260 DISP "Hfe(max)=1E? (1E3)"
1270 INPUT H2
1280 M1=1.E-15
1290 M2=.01
1300 GCLEAR
1310 GRAPH
1320 SCALE LGT(M1),LGT(M2),LGT(H
     1),LGT(H2)
1330 AXES 1,1,LGT(M1),LGT(H1)
1340 MOVE LGT(M2)+(LGT(M2)-LGT(M
     1))*.02,LGT(H1)+(LGT(H2)-LG
     T(H1))*.1
1350 LDIR 0
1360 LORG 5
1370 CSIZE 4
1380 LABEL "Ic(A)"
1390 MOVE LGT(M1)-(LGT(M2)-LGT(M
     1))*.2,LGT(H2)*(1/5)
1400 LDIR 90
1410 LABEL "Hfe"
1420 MOVE (LGT(M2)+LGT(M1))/2,1.
     2*LGT(H2)
1430 LDIR 0
1440 LABEL "Hfe-Ic CHARACTERISTI
     CS"
1450 MOVE (LGT(M2)+LGT(M1))/2,1.
     05*LGT(H2)
1460 LABEL "HP 4140B"
1470 CSIZE 3
1480 LDIR 90
1490 LORG 8
1500 FOR N=LGT(M1) TO LGT(M2) ST
     EP 1
1510 MOVE N,LGT(H1)-(LGT(H2)-LGT
     (H1))*.01
1520 LABEL USING 150 ; 10^N
1530 NEXT N
1540 N=0
1550 MOVE LGT(M1)-(LGT(M2)-LGT(M
     1))*.01,LGT(H1*10^N)
1560 LDIR 0
1570 LORG 8
1580 LABEL H1*10^N
1590 N=N+1
1600 IF N>LGT(H2)-LGT(H1) THEN 1
     620
1610 GOTO 1550
1620 ! *** Hfe CALCULATION ***
1630 FOR I=1 TO N2
1640 PEN SGN(A(I)/C(I))
1650 PLOT LGT(ABS(A(I))),LGT(ABS
     (A(I)/C(I)))
1660 NEXT I
1670 COPY
1680 RETURN
```

APPENDIX B: "g_m-V_{GS}" Program Listing for the HP85

For DUT connection, refer to Figure 4.

```
 10 ! *** Gm-Vgs MEASUREMENT ***
 20 ! ..."Gm-Vgs"....MAR.12,1981
 30 M=717
 40 OPTION BASE 1
 50 DIM S$[30],D$[16]
 60 DIM I1(100),G1(100),V(100)
 70 DISP "4140B Address Code ? (
     717)"
 80 INPUT M
 90 PRINTER IS 16
100 GOSUB 220 ! Meas-setting
110 GOSUB 350 ! Measurement
120 GOSUB 480
130 GOSUB 1100 ! Gm-calculation
140 GOSUB 1200 ! Graphic-output
150 END
160 !
170 IMAGE 2A,M3DZ.2D,","
180 IMAGE 18A,": ",MDCDDZ.DD,K
190 IMAGE MZ.DD," V  ",MZ.DDDe,"
     A"
200 IMAGE 18A,": ",12A
210 !
220 ! *** Meas-setting ***
230 DISP "DUT=? (Up to 10 charac
     ters)"
240 INPUT S$
250 DISP "DATE=?"
260 INPUT D$
270 DISP "Input ; START V?, STOP
     V?, STEP V?, STEP DELAY(sec
     )?, HOLD TIME(sec)?,Vds"
280 INPUT V1,V2,V3,T1,T2,V0
290 N1=ABS((V2-V1)/V3)
300 IF FP(N1)#0 THEN N2=INT(N1)+
     2
310 IF FP(N1)=0 THEN N2=INT(N1)+
     1
320 REDIM I1(N2),G1(N2),V(N2)
330 RETURN
340 !
350 ! *** Measurement ***
360 DISP "Connect DUT for Id-Vgs
     Meas. as D:HI, G:Va, S:GND,
     LO:Vb 'CONT'"
370 PAUSE
380 OUTPUT M ;"F2RA1I1J1T2A3B1L3
     M3"
390 OUTPUT M USING 170 ; "PS",V1
     ,"PT",V2,"PE",V3,"PD",T1,"PH
     ",T2,"PB",V0
400 OUTPUT M ;"W1"
410 DISP "MEASURING"
420 OUTPUT M ;"W1"
430 FOR I=1 TO N2
440 ENTER M ; I1(I),V(I)
450 DISP USING 190 ; V(I),I1(I)
460 NEXT I
470 RETURN
480   *** Graphic-output ***
490 GCLEAR
500 GRAPH
510 LOCATE 30,110,25,88
520 PEN 1
530 M1=1.E-15
```

```
540 M2=.01
550 SCALE V1,V2,LGT(M1),LGT(M2)
560 AXES 1,1,V1,LGT(M1)
570 DEG
580 X=V2-V1
590 Y=LGT(M2)-LGT(M1)
600 Y1=LGT(M1)
610 Y2=LGT(M2)
620 MOVE V1+X*.55,Y1+Y*1.15
630 LORG 5
640 LABEL "Id-Vgs Characteristic
    s"
650 MOVE V1+X*.25,Y1+Y*1.07
660 FXD 2
670 LABEL "DUT=";S$
680 MOVE V1+.9*X,Y1+Y*1.05
690 IMAGE "Vds=",MZ.DD,"V"
700 LABEL USING 690 ; V0
710 MOVE V1-X*.3,Y1+Y*.55
720 LDIR 90
730 LABEL "Id (A)"
740 LDIR 0
750 MOVE V1+X*.31,Y1-Y*.2
760 LABEL "HP 4140B"
770 MOVE V1+X*.99,Y1-Y*.2
780 LABEL "Vgs (V)"
790 MOVE V1+X*.99,Y1-Y*.3
800 LABEL D$
810 PEN 1
820 ! *** Vgs Axis unit plot ***
830 CSIZE 3
840 LDIR 0
850 LORG 6
860 FOR I=INT(V1) TO INT(V2)
870 MOVE I,Y1-.1
880 FXD 1
890 LABEL I
900 NEXT I
910 ! *** Id Axis unit plot ***
920 CSIZE 3
930 LDIR 0
940 LORG 8
950 FOR I=Y1 TO Y2
960 MOVE V1-X*.01,I
970 LABEL USING "MZe" ; 10^I
980 NEXT I
990 PEN 1
1000 LINETYPE 1
1010 FOR I=1 TO N2
1020 PLOT V1+(I*V3-V3),LGT(ABS(I
     1(I)))
1030 IF V1+(I*V3-V3)>V2 THEN 105
     0
1040 NEXT I
1050 PENUP
1060 COPY
1070 DISP "PRESS 'CONT'"
1080 PAUSE
1090 RETURN
1100 ! *** Gm-calculate ***
1110 FOR I=1 TO N2-1
1120 G1(I)=ABS((I1(I+1)-I1(I))/V
     3)
1130 NEXT I
1140 G2=0
1150 FOR I=1 TO N2-1
1160 G2=MAX(G2,G1(I))
1170 NEXT I
1180 G3=10^(INT(LGT(G2))-(FP(LGT
     (G2))#0 AND LGT(G2)>=0))
1190 RETURN
1200 ! *** Graphic-output ***
1210 GCLEAR
1220 GRAPH
1230 Y=G3*(1+INT(G2/G3))
1240 Y1=0
1250 SCALE V1,V2,0,G3*(1+INT(G2/
     G3))
1260 AXES 1,G3,INT(V1),0
1270 ! *** Gm unit x plot ***
1280 CSIZE 3
1290 LDIR 0
1300 LORG 6
1310 FOR I=INT(V1) TO INT(V2)
1320 MOVE I,-G2*.01
1330 FXD 0
1340 LABEL I
1350 NEXT I
1360 ! *** Gm unit y plot ***
1370 CSIZE 3
1380 LDIR 0
1390 LORG 8
1400 FOR I=0 TO INT(G2/G3)+(FP(L
     GT(G2))#0)
1410 MOVE INT(V1)-(INT(V2)-INT(V
     1))*.01,I*G3
1420 FXD 0
1430 LABEL I
1440 NEXT I
1450 ! *** Gm PLOT ***
1460 MOVE V1+X*.55,Y1+Y*1.15
1470 LORG 5
1480 LABEL "Gm-Vgs Characteristi
     cs"
1490 MOVE V1+X*.25,Y1+Y*1.07
1500 FXD 2
1510 LABEL "DUT=";S$
1520 MOVE V1+X,Y1+Y*1.05
1530 IMAGE "Gm (S) *",MZe
1540 LABEL USING 690 ; V0
1550 MOVE V1-X*.15,Y1+Y*.55
1560 LDIR 90
1570 LABEL USING 1530 ; G3
1580 LDIR 0
1590 MOVE V1+X*.31,Y1-Y*.2
1600 LABEL "HP 4140B"
1610 MOVE V1+X*.99,Y1-Y*.2
1620 LABEL "Vgs (V)"
1630 MOVE V1+X*.99,Y1-Y*.3
1640 LABEL D$
1650 PEN 1
1660 LINETYPE 1
1670 FOR I=1 TO N2-1
1680 PLOT V1+(I-1)*V3,G1(I)
1690 IF V1+(I-1)*V3>V2 THEN 150
1700 NEXT I
1710 COPY
1720 RETURN
```

HEWLETT PACKARD

For more information, call your local HP Sales Office or nearest Regional Office: • Eastern (201) 265-5000; • Midwestern (312) 255-9800; • Southern (404) 955-1500; • Western (213) 970-7500; • Canadian (416) 678-9430. Ask the operator for instrument sales. Or write Hewlett-Packard, 1501 Page Mill Road, Palo Alto, CA 94304. **In Europe:** Hewlett-Packard S.A., 7, rue du Bois-du-Lan, P.O. Box, CH 1217 Meyrin 2, Geneva, Switzerland. **In Japan:** Yokogawa-Hewlett-Packard Ltd., 29-21, Takaido-Higashi 3-chome, Suginami-ku, Tokyo 168.

Printed in JAPAN. APRIL, 1981 5952-8873